

### Problema del Trasporto

Una ditta di trasporto deve trasferire container vuoti dai propri 6 Magazzini, situati a Verona, Perugia, Roma, Pescara, Taranto e Lamezia, ai principali Porti nazionali (Genova, Venezia, Ancona, Napoli, Bari).

Le disponibilità di container vuoti ai Magazzini sono le seguenti

	Container vuoti
Verona	10
Perugia	12
Roma	20
Pescara	24
Taranto	18
Lamezia	40

e le richieste ai Porti sono le seguenti:

	Container richiesti
Genova	20
Venezia	15
Ancona	25
Napoli	33
Bari	21

Trasportare i container dai magazzini ai porti costa. Il costo di trasporto di ciascun container è proporzionale alla distanza percorsa dal camion che lo trasporta. Il costo di trasporto chilometrico per ogni container è di 300 lire / km. Le distanze fra Magazzini e Porti sono riportate nella seguente tabella:

	Genova	Venezia	Ancona	Napoli	Bari
Verona	290 km	115 km	355 km	715 km	810 km
Perugia	380 km	340 km	165 km	380 km	610 km
Roma	505 km	530 km	285 km	220 km	450 km
Pescara	655 km	450 km	155 km	240 km	315 km
Taranto	1010 km	840 km	550 km	305 km	95 km
Lamezia	1072 km	1097 km	747 km	372 km	333 km

1. Si vuole determinare la politica di trasporto di costo complessivo minimo, formulando un modello di PM e risolvendolo con AMPL.
2. Ci sono dei porti con delle richieste non soddisfatte? Se sì, quali?
3. Di quanto cambierebbe la funzione obiettivo se la disponibilità di container vuoti a Perugia aumentasse di 2 unità?

4. Di quanto può variare la disponibilità di container vuoti al magazzino di Lamezia senza che la soluzione cambi?
5. Nella formulazione, è necessario imporre l'integralità delle variabili come vincolo al problema?

### Soluzione

1. Prima di tutto si noti che il problema è di PLI perché si parla di numero di container. Tuttavia approssimeremo il problema con un rilassamento continuo.

- FORMULAZIONE.

- *Indici:*

- \* sia  $i$  un indice sull'insieme  $\{1, \dots, M\}$  corrispondente ai magazzini ( $M = 6$ : 1=Verona, 2=Perugia, 3=Roma, 4=Pescara, 5=Taranto, 6=Lamezia);
- \* sia  $j$  un indice sull'insieme  $\{1, \dots, P\}$  corrispondente ai porti ( $P = 5$ : 1=Genova, 2=Venezia, 3=Ancona, 4=Napoli, 5=Bari).

- *Parametri:*

- \* sia  $m_i$  la disponibilità (in numero di container) dell' $i$ -esimo magazzino ( $m_1 = 10, m_2 = 12, m_3 = 20, m_4 = 24, m_5 = 18, m_6 = 40$ );
- \* sia  $r_j$  la richiesta di container al  $j$ -esimo porto ( $r_1 = 20, r_2 = 15, r_3 = 25, r_4 = 33, r_5 = 21$ ).
- \* sia  $d_{ij}$  la distanza tra il magazzino  $i$  e il porto  $j$  (dati come da tabelle nel testo del problema);
- \* sia  $C$  il costo di trasporto unitario al chilometro ( $C = 300$ ).

- *Variabili:*

- \* sia  $x_{ij}$  il numero di container spediti dal magazzino  $i$  al porto  $j$ ;

Tutte le variabili sono non negative.

- *Funzione Obiettivo:*

$$\min \sum_{i=1}^M \sum_{j=1}^P C d_{ij} x_{ij}$$

- *Vincoli:*

- (a) (disponibilità magazz.) per ogni  $i \leq M$ :  $\sum_{j=1}^P x_{ij} \leq m_i$ ;
- (b) (richiesta porti) per ogni  $j \leq P$ :  $\sum_{i=1}^M x_{ij} \geq r_j$ .

- FILE MODELLO DI AMPL: trasporto.mod.

```
# trasporto.mod - file modello per AMPL - problema del trasporto
set MAGAZZINI;
set PORTI;
```

```
param capienza { MAGAZZINI } >= 0;
```

```

param richiesta { PORTI } >= 0;
param distanza { MAGAZZINI, PORTI } >= 0;
param costokm >= 0;

var x { MAGAZZINI, PORTI } >= 0;

minimize costo:
    sum {i in MAGAZZINI, j in PORTI}
        costokm * distanza[i,j] * x[i,j];

subject to disponibilita {i in MAGAZZINI}:
    sum {j in PORTI} x[i,j] <= capienza[i];

subject to domanda {j in PORTI}:
    sum {i in MAGAZZINI} x[i,j] >= richiesta[j];

```

- FILE DATI DI AMPL: trasporto.dat.

```

# trasporto.mod - file modello per AMPL - problema del trasporto
set MAGAZZINI := Verona Perugia Roma Pescara Taranto Lamezia;
set PORTI := Genova Venezia Ancona Napoli Bari;
param capienza :=
    Verona 10
    Perugia 12
    Roma 20
    Pescara 24
    Taranto 18
    Lamezia 40 ;
param richiesta :=
    Genova 20
    Venezia 15
    Ancona 25
    Napoli 33
    Bari 21 ;
param distanza :
    Genova Venezia Ancona Napoli Bari :=
    Verona 290 115 355 715 810
    Perugia 380 340 165 380 610
    Roma 505 530 285 220 450
    Pescara 655 450 155 240 315
    Taranto 1010 840 550 305 95
    Lamezia 1072 1097 747 372 333 ;
param costokm := 300;

```

- SOLUZIONE. I comandi da dare ad AMPL per ottenere la soluzione sono

```

model trasporto.mod;
data trasporto.dat;
option solver cplex;
solve;
option display_round 4;
display costo;
display x;

```

La soluzione è

```

costo = 9045900.0000
x [*,*]
:      Ancona      Bari      Genova      Napoli      Venezia      :=
Lamezia  0.0000      3.0000      0.0000      27.0000      0.0000
Perugia  0.0000      0.0000      7.0000      0.0000      5.0000
Pescara  24.0000      0.0000      0.0000      0.0000      0.0000
Roma     1.0000      0.0000      13.0000     6.0000      0.0000
Taranto  0.0000      18.0000     0.0000      0.0000      0.0000
Verona   0.0000      0.0000      0.0000      0.0000      10.0000 ;

```

2. Si richiede ad AMPL di mostrare il valore delle variabili slack associate ai vincoli di richiesta nei porti:

```
display domanda.slack;
```

Il risultato è che non ci sono richieste insoddisfatte in nessun porto.

3. Si richiede ad AMPL di mostrare il valore dei prezzi ombra corrispondenti ai vincoli di disponibilità in magazzino:

```
display disponibilita;
```

Risulta che la variazione nel valore della funzione obiettivo corrispondente a un incremento unitario della capacità del magazzino di Perugia è di -83100. Dunque, aumentandone la capacità di 2 unità si ottiene una diminuzione di costi di 166200.

4. Si richiede ad AMPL di mostrare il valore delle variabili di slack dei vincoli di disponibilità:

```
display disponibilita.slack;
```

Risulta che la variabile di slack del vincolo di disponibilità per Lamezia è 10, perciò nel magazzino di Lamezia avanzano 10 container e quindi affinché la soluzione non cambi la disponibilità di container può o diminuire al massimo di 10 unità o aumentare di una qualunque quantità.

5. Per risolvere un modello preciso del problema (cioè, non il rilassamento continuo), è necessario aggiungere la parola-chiave `integer` ai limiti delle variabili, ovvero

```
var x { MAGAZZINI, PORTI } >= 0, integer;
```

La soluzione data da AMPL è la stessa che per il rilassamento continuo. Questo succede perché la matrice  $A$  dei vincoli del problema  $Ax = b$  è totalmente unimodulare.